Université PARIS-XII — Faculté de Sciences économiques et de gestion
Master *Sciences économiques*
Mention *Économie quantitative et expertise internationale*
Économétrie III — *F. LEGENDRE* — année universitaire 2005-06

Fiche de travail n°2
Mes premières estimations

Je vous propose dans cette deuxième fiche de travail d'estimer une fonction de production en utilisant les données collectées dans la fiche de travail n°1.

1 Estimer la fonction de production en niveau

Dans un premier temps, la fonction de production est estimée «en niveau». On va ajuster le modèle suivant :

$$(1) \quad q_{it} = \alpha k_{it} + \beta \ell_{it} + \gamma t + \delta + u_{it}$$

où q est le logarithme de la valeur ajoutée en volume, k le logarithme du stock de capital net en volume, ℓ le logarithme de l'emploi total, t le trend temporel et u le terme d'erreurs. L'indice i repère la branche ; l'indice t l'année.

Donner les valeurs que l'on doit attendre pour α, β et γ. Peut-on *a priori* donner une valeur pour le coefficient δ ?

Le modèle peut aussi être estimé en introduisant des constantes individuelles :

$$(2) \quad q_{it} = \alpha k_{it} + \beta \ell_{it} + \gamma t + \delta_i + u_{it}$$

Enfin, il n'est pas déraisonnable d'imaginer que le rythme du progrès technique ne soit pas le même dans chaque branche. On estimerait alors le modèle suivant :

$$(3) \quad q_{it} = \alpha k_{it} + \beta \ell_{it} + \gamma_i t + \delta_i + u_{it}$$

Le programme SAS suivant estime ces différents modèles.

```
DATA table ;
  SET table ;
  lq = LOG(q) ;
  ll = LOG(l) ;
  lk = LOG(k) ;
  ib01 = branche EQ 1 ;
...
  ib13 = branche EQ 13 ;
  itb01 = ib01*t ;
...
  itb13 = ib13*t ;
RUN ;
PROC reg DATA = table ;
  MODEL lq = lk ll t ;
  MODEL lq = lk ll t ib01-ib13 / NOINT ;
  MODEL lq = lk ll itb01-itb13 ib01-ib13 / NOINT ;
RUN ;
PROC glm DATA = table ;
  CLASS branche ;
  MODEL lq = lk ll t branche / SOLUTION ;
RUN ;
PROC glm DATA = table ;
  CLASS branche ;
  MODEL lq = lk ll branche*t branche / SOLUTION ;
RUN ;
```

Les indicatrices de branche sont engendrées explicitement au moyen d'une expression logique. Par exemple, l'expression «`branche EQ 1`» est égale à *oui* (c'est-à-dire à `1`) si le code de la branche est égal à 1. Notez la façon dont sont engendrées les indicatrices de trend temporel par branche. Enfin, la `PROC glm` est en mesure d'estimer ces mêmes modèles sans qu'il soit nécessaire de définir explicitement les variables indicatrices ; il suffit, au moyen de l'instruction `CLASS`, de déclarer les variables qui repèrent les différents groupes.

2 Estimer la fonction de production en taux de croissance

On peut aussi vouloir estimer la fonction de production «en taux de croissance». On ajusterait alors les deux modèles suivants :

$$(4) \quad \Delta q_{it} = \alpha \Delta k_{it} + \beta \Delta \ell_{it} + \gamma + u_{it}$$

$$(5) \quad \Delta q_{it} = \alpha \Delta k_{it} + \beta \Delta \ell_{it} + \gamma_i + u_{it}$$

où l'opérateur Δ est l'opérateur de différence au cours du temps : $\Delta z_{it} = z_{it} - z_{i\,t-1}$. Notez que $z_{it} - z_{i\,t-1}$, si z est le logarithme de Z, est approximativement égal au taux de croissance de Z_{it}.

Pour engendrer les variables, le programme SAS est :

```
DATA table ;
  SET table ;
    BY branche ;
  dlq = lq - LAG(lq) ;
  dlk = lk - LAG(lk) ;
  dll = ll - LAG(ll) ;
  IF FIRST.branche THEN DO ;
    dlq = . ; dlk = . ; dll = . ;
  END ;
RUN ;
```

Notez que l'instruction `SET` est accompagnée d'une directive `BY` : on dispose ainsi de l'expression logique `FIRST.branche` qui est égale à *oui* si l'itération courante de l'étape `DATA` est relative à la première observation d'une branche. On peut de la sorte définir à «manquant» les taux de croissance.